
                                                                                     
_b_e_g_i_n
   _c_o_m_m_e_n_t GIER DEMONSTRATION PROGRAM−1D;
   _b_o_o_l_e_a_n newnim, newmap, newlineq, newprime, newlanu;
   _i_n_t_e_g_e_r linerest, oldrand, type;
   _s_w_i_t_c_h TYPE := NIM, MAP, LINEQ, PRIME, LANU, FINISH;
   _p_r_o_c_e_d_u_r_e NEWPAGE;
   _b_e_g_i_n
      _f_o_r linerest := linerest − 1 _w_h_i_l_e linerest _> −9 _d_o skrvvr;
      linerest := 62
   _e_n_d NEWPAGE;
   _p_r_o_c_e_d_u_r_e LINE;
   _b_e_g_i_n
      linerest := linerest − 1;
      skrvvr;
      _i_f linerest < 0 _t_h_e_n NEWPAGE
   _e_n_d LINE;
   _p_r_o_c_e_d_u_r_e SHIFT(n);
   _v_a_l_u_e n;
   _i_n_t_e_g_e_r n;
   _i_f linerest < n _t_h_e_n NEWPAGE;
   _p_r_o_c_e_d_u_r_e CHECKLINE;
   _i_f tegn= 64 ∨  tegn= 192 _t_h_e_n
   _b_e_g_i_n
      linerest := linerest − 1;
      SHIFT(0)
   _e_n_d CHECKLINE;
   _i_n_t_e_g_e_r _p_r_o_c_e_d_u_r_e RANDOM(n);
   _v_a_l_u_e n;
   _i_n_t_e_g_e_r n;
   _b_e_g_i_n
      _r_e_a_l y, MOD;
      MOD := 32768;
      y := oldrand ×6859;
      oldrand := y − MOD ×entier(y/MOD);
      RANDOM := 1 + entier(n ×oldrand/MOD)
   _e_n_d RANDOM;
START: oldrand := 999;
   linerest := 65;
   newnim := newmap := newlineq := newprime := newlanu := _t_r_u_e;
   skrvtegn(62);
   skrvtekst(|<<
                          GIER DEMONSTRATION PROGRAM 1D

     Programmet kan bruges paa 5 forskellige maader:

         1.   Tændstikspillet: NIM.
         2.   Trykning af tilfældige landkort.
         3.   Løsning af tilfældige lineære ligninger.
         4.   Beregning af primtal.
         5.   Beregning af store tal.
              6 giver program slut.
     Skriv Deres initialer her:|>);
   linerest := linerest − 11;
   _b_e_g_i_n
      _i_n_t_e_g_e_r i, j, sum;
      sum := tasttegn+ tasttegn;
      _f_o_r i := 1 _s_t_e_p 1 _u_n_t_i_l sum _d_o j := RANDOM(1)
   _e_n_d of advance of random procedure;
RESTART: LINE;  LINE;
   SHIFT(10);
   skrvtekst(|<<Vælg program type 1−5 (6 giver program stop): |>);
   type := tasttegn;
   LINE;  LINE;
   _g_o_ _t_o TYPE[type];
   _g_o_ _t_o RESTART;
NIM: SHIFT(10);



   skrvtekst(|<<Type 1.  Tændstikspillet:  NIM|>);
   LINE;  LINE;
   _b_e_g_i_n _c_o_m_m_e_n_t NIM−block;
      _b_o_o_l_e_a_n longtext, winmessage, wrongmessage, loosemessage, present;
      _i_n_t_e_g_e_r M, G, N, g, n, t, gno, remove, fact, R, boolsum, aritsum, ask,
              GMAX, nmax;
      longtext := newnim;
      _i_f newnim _t_h_e_n
      _b_e_g_i_n
         SHIFT(10);
         skrvtekst(|<<
                             SPILLEREGLER FOR NIM
     Spillet begynder med et tilfældigt udvalg af G bunker af tændstikker.  Hver
bunke indeholder højst M tændstikker.  M skrives som 2| ∧ N − 1, og De skal opgive
N og G.  Vi vil saa skiftevis fjerne tændstikker fra bunkerne.  Den, som
fjerner den eller de sidste tændstikker, har vundet.  Kun een bunke maa røres
i hvert træk, og man skal fjerne mindst een tændstik fra den bunke.
|>);
         linerest := linerest − 7
      _e_n_d if newnim;
AGAIN: SHIFT(4);
      LINE; 
      winmessage := wrongmessage := loose message :=  _f_a_l_s_e;
      skrvtekst(|<<Opgiv N: |>);
      N := tast;
      _i_f N > 9 _t_h_e_n N := 9;
      LINE;
      CHECKLINE;
      M := 2| ∧ N − 1;
      ask := 0;
      skrvtekst(|<<Opgiv G: |>);
      G := tast;
      _i_f G > 15 _t_h_e_n G := 15;
      LINE;
      CHECKLINE;
      LINE;  LINE;
      _b_e_g_i_n _c_o_m_m_e_n_t inner NIM−block;
         _i_n_t_e_g_e_r _a_r_r_a_y GROUP[1:G], SUM[1:N], BITS[1:G, 1:N];
         _p_r_o_c_e_d_u_r_e PRINTGROUPS;
         _f_o_r g := 1 _s_t_e_p 1 _u_n_t_i_l G _d_o skrv(|<−ndd|>, GROUP[g]);
         _p_r_o_c_e_d_u_r_e DISPLAY BITS(g);
         _v_a_l_u_e g;
         _i_n_t_e_g_e_r g;
         _b_e_g_i_n
            R := GROUP[g];
            fact := 2;
            _f_o_r n := 1 _s_t_e_p 1 _u_n_t_i_l N _d_o
            _b_e_g_i_n
               present := R _: fact ×fact |= R;
               _i_f present _t_h_e_n R := R − fact _: 2;
               BITS[g,n] := _i_f present _t_h_e_n 1 _e_l_s_e 0;
               fact := fact ×2
            _e_n_d for n
         _e_n_d DISPLAY BITS;
         _p_r_o_c_e_d_u_r_e FIND SUM;
         _b_e_g_i_n
            aritsum := 0;
            _f_o_r g := 1 _s_t_e_p 1 _u_n_t_i_l G _d_o aritsum := aritsum + GROUP[g];
            _f_o_r n := 1 _s_t_e_p 1 _u_n_t_i_l N _d_o
            _b_e_g_i_n
               boolsum := 0;
               _f_o_r g := 1 _s_t_e_p 1 _u_n_t_i_l G _d_o boolsum := boolsum + BITS[g,n];
               SUM[n] := boolsum − boolsum _: 2 ×2
            _e_n_d;
            boolsum := 0;  
            fact := 1;
            _f_o_r n := 1 _s_t_e_p 1 _u_n_t_i_l N _d_o



            _b_e_g_i_n
               boolsum := boolsum + SUM[n] ×fact;
               fact := 2 ×fact
            _e_n_d for n
         _e_n_d FIND SUM;
         _f_o_r g := 1 _s_t_e_p 1 _u_n_t_i_l G _d_o
         _b_e_g_i_n
            GROUP[g] := RANDOM(M);
            DISPLAY BITS(g)
         _e_n_d for g;
         FIND SUM;
         SHIFT(4);
         skrvtekst(|<<Her er bunkerne:|>);
         LINE;
         skrvtekst(|<<Bunke nr.:        |>);
         _f_o_r g := 1 _s_t_e_p 1 _u_n_t_i_l G _d_o skrv(|<−ndd|>, g);
         LINE;
         skrvtekst(|<<Antal tændstikker:|>);
         PRINTGROUPS;
BB:      SHIFT(4);
         skrvtekst(|<<
     Hvis De ønsker at gøre det første træk, skriver De et 1−tal her, ellers
et 2−tal: |>);
         linerest := linerest − 2;
         t := tasttegn;
         _i_f t |= 1 ∧  t |= 2 _t_h_e_n _g_o_ _t_o BB;
         _i_f t = 2 _t_h_e_n _g_o_ _t_o II;
GG:      SHIFT(4);
         LINE;
         skrvtekst(_i_f longtext _t_h_e_n
|<<Skriv her nummeret paa den bunke, fra hvilken De vil fjerne tændstikker: |>
         _e_l_s_e |<<Vælg Deres bunke: |>);
CC:      gno := tast;
         CHECKLINE;
         LINE;
         _i_f gno < 1 ∨  gno > G _t_h_e_n
         _b_e_g_i_n
            skrvtekst(
|<<Undskyld, men tallet er for |>, _i_f gno < 1 _t_h_e_n |<<lille.|> _e_l_s_e |<<stort.|>);
DD:         LINE;
            skrvtekst(|<<  Prøv igen her: |>);
            _g_o_ _t_o CC
         _e_n_d if out of range;
         _i_f GROUP[gno] = 0 _t_h_e_n
         _b_e_g_i_n
            skrvtekst(|<<Undskyld, men denne bunke er tom.|>);
            _g_o_ _t_o DD
         _e_n_d if empty group;
         remove := GROUP[gno];
         _i_f remove |= 1 _t_h_e_n
         _b_e_g_i_n
            SHIFT(4);
            skrvtekst(_i_f longtext _t_h_e_n
|<<Og antallet af tændstikker, De vil fjerne: |> _e_l_s_e |<<Og antallet: |>);
EE:         remove := tast;
            CHECKLINE;
            LINE;
            _i_f remove < 1 _t_h_e_n
            _b_e_g_i_n
               skrvtekst(|<<De skal fjerne mindst een tændstik.  Prøv igen: |>);
               _g_o_ _t_o EE
            _e_n_d;
            _i_f remove > GROUP[gno] _t_h_e_n
            _b_e_g_i_n
               skrvtekst(
|<<Saa mange er der ikke i bunken.  De fjerner altsaa hele bunken.|>);
               LINE;



               remove := GROUP[gno]
            _e_n_d if too many
         _e_n_d if more than one;
         GROUP[gno] := GROUP[gno] − remove;
         _i_f longtext _t_h_e_n longtext := _f_a_l_s_e;
         DISPLAY BITS(gno);
         FIND SUM;
II:      _i_f boolsum |= 0 _t_h_e_n
         _b_e_g_i_n
            SHIFT(4);
            _i_f winmessage ∧  −, wrongmessage _t_h_e_n
            _b_e_g_i_n
               skrvtegn(29);
               skrvtekst(|<<Det var forkert.  Nu kan De ikke vinde.|>);
FF:            loosemessage := wrongmessage := _t_r_u_e;
               skrvtegn(62);
               LINE
            _e_n_d of blunder;
            _i_f loosemessage _t_h_e_n
            _b_e_g_i_n
               ask := ask + 1;
               _i_f ask _: 3 ×3 = ask _t_h_e_n
               _b_e_g_i_n
                  SHIFT(4);
                  skrvtekst(
|<<Hvis De giver fortabt, saa skriv et 1−tal her: |>);
                  t := tasttegn;
                  LINE;
                  _i_f t = 1 _t_h_e_n _g_o_ _t_o ASK FOR MORE
               _e_n_d if third time;
               _g_o_ _t_o GIERMOVE
            _e_n_d;
            skrvtegn(29);
            skrvtekst(|<<De kan ikke vinde dette spil.|>);
            _g_o_ _t_o FF
         _e_n_d if boolsum |= 0;
         _i_f aritsum = 0 _t_h_e_n
         _b_e_g_i_n
            SHIFT(4);
            skrvtekst(|<<De har vundet.  Tillykke.|>);
            _g_o_ _t_o ASK FOR MORE
         _e_n_d if finished;
         _i_f −, winmessage _t_h_e_n
         _b_e_g_i_n
            SHIFT(4);
            winmessage := _t_r_u_e;
            skrvtegn(29);
            skrvtekst(|<<Hvis De spiller rigtigt, kan De vinde dette spil.|>);
            skrvtegn(62);
            LINE
         _e_n_d if not winmessage;
         GMAX := GROUP[1];
         gno := 1;
         _f_o_r g := 2 _s_t_e_p 1 _u_n_t_i_l G _d_o
         _b_e_g_i_n
            _i_f GROUP[g] > GMAX _t_h_e_n
            _b_e_g_i_n
               GMAX := GROUP[g];
               gno := g
            _e_n_d
         _e_n_d search of largest group;
         remove := 1;
JJ:      GROUP[gno] := GROUP[gno] − remove;  
         SHIFT(4);
         _i_f GROUP[gno] > 0 _t_h_e_n
         _b_e_g_i_n
            skrvtekst(|<<Jeg fjerner nu|>);



            skrv(|<−ndd|>, remove);
            skrvtekst(|<< fra bunke nr.|>)
         _e_n_d group not empty
         _e_l_s_e
         skrvtekst(|<<Jeg fjerner nu hele bunke nr.|>);
         skrv(|<−ndd|>, gno);
         skrvtekst(|<<. Bunkerne indeholder nu:|>);
         LINE;
         LINE;
         PRINTGROUPS;
         DISPLAY BITS (gno);
         FIND SUM;
         _i_f aritsum > 0 _t_h_e_n _g_o_ _t_o GG;
         SHIFT(4); 
         LINE;
         skrvtekst(|<<De har tabt.|>);
ASK FOR MORE:  LINE;
         newnim := _f_a_l_s_e;
HH:      skrvtekst(|<<
Hvis De ønsker at prøve igen, saa skriv et 1−tal her, ellers et 2−tal: |>);
         linerest := linerest − 1;
         t := tasttegn;
         _i_f t |= 1 ∧  t |= 2 _t_h_e_n _g_o_ _t_o HH;
         LINE;  LINE;
         _g_o_ _t_o _i_f t = 1 _t_h_e_n AGAIN _e_l_s_e RESTART;
GIERMOVE: _f_o_r g := 1 _s_t_e_p 1 _u_n_t_i_l G _d_o
         _b_e_g_i_n
            _i_f boolsum = GROUP[g] _t_h_e_n
            _b_e_g_i_n
               remove := boolsum;
               gno := g;
               _g_o_ _t_o JJ
            _e_n_d if remove whole group
         _e_n_d for g;
         _f_o_r n := N _s_t_e_p −1 _u_n_t_i_l 1 _d_o
         _b_e_g_i_n
            _i_f SUM[n] = 1 _t_h_e_n
            _b_e_g_i_n
               nmax := n;
               _g_o_ _t_o KK
            _e_n_d hit
         _e_n_d for n;
KK:      _f_o_r g := 1 _s_t_e_p 1 _u_n_t_i_l G _d_o
         _b_e_g_i_n
            _i_f BITS[g, nmax] = 1 _t_h_e_n
            _b_e_g_i_n
               gno := g;
               remove := 0;
               fact := 1;
               _f_o_r n := 1 _s_t_e_p 1 _u_n_t_i_l nmax _d_o
               _b_e_g_i_n
                  _i_f SUM[n] = 1 _t_h_e_n
                  remove := remove + (_i_f BITS[gno, n] = 1 _t_h_e_n fact _e_l_s_e − fact);
                  fact := fact ×2
               _e_n_d for n;
               _g_o_ _t_o JJ
            _e_n_d if hit Bits
         _e_n_d for g;
         _g_o_ _t_o HH
      _e_n_d inner NIM−block
   _e_n_d NIM;
MAP:
   _b_e_g_i_n _c_o_m_m_e_n_t MAP−block;
      _b_o_o_l_e_a_n red, even;
      _i_n_t_e_g_e_r a, amin, amax, b, bs, fh, fhmin, fhmax, fv, fvmin, fvmax, h, hs,
              k, n, ncon, nmap, p, p2, p3, q, r, r1, r2, s, s1, s2, spr;
      _r_e_a_l d, dmin, e, f, g, j, v;



      SHIFT(10);
      skrvtekst(|<<Type 2.  Trykning af tilfældige landkort.|>);
      LINE;  LINE;
      _i_f newmap _t_h_e_n
      _b_e_g_i_n
         SHIFT(18);
         skrvtekst(|<<
     Følgende parametre bruges i dette program:
                                                   Typiske værdier:
     nmap:    Antal landkort.                             1
     h:       Kortenes højde.                            60
     b:       Kortenes bredde.                           80
     hs:      Højde af delkort.                          15
     bs:      Bredde af delkort.                         20
     ncon:    Antal lande per delkort.                    1
     fh:      Vandret skalafaktor.                        1−5
              Minimum:  fhmin,  maximum:  fhmax.
     fv:      Lodret skalafaktor.                         1−5
              Minimum:  fvmin,  maximum:  fvmax.
     a:       Drejningsvinkel.                           45
              Minimum:  amin,  maximum:  amax.
     spr:     Spredningsfaktor:  Minimum  1, maximum  10  1
|>);
         linerest := linerest − 16;
         newmap := _f_a_l_s_e
      _e_n_d if newmap;
LL:   LINE;
      SHIFT(5);
      skrvtekst(|<<
     Opgiv parametrene:
 nmap  h   b   hs  bs  ncon  fhmin fhmax fvmin fvmax amin amax  spr
|>);
      nmap := tast;  h := tast;  b := tast;
      hs := tast;  bs := tast;  ncon := tast;
      fhmin := tast;  fhmax := tast;  fvmin := tast;
      fvmax := tast;  amin := tast;  amax := tast;
      spr := tast;
      r1 := h _: hs;
      s1 := b _: bs;
      p2 := r1 ×s1 ×ncon;
      linerest := linerest − 3;
      LINE;
      _f_o_r n := 1 _s_t_e_p 1 _u_n_t_i_l nmap _d_o
      _b_e_g_i_n _c_o_m_m_e_n_t inner MAP−block;
         _i_n_t_e_g_e_r _a_r_r_a_y cx, cy, fh, fv[1:p2];
         _a_r_r_a_y cosv, sinv[1:p2];
         p := 0;
         _f_o_r r := 1 _s_t_e_p 1 _u_n_t_i_l r1 _d_o
         _f_o_r s := 1 _s_t_e_p 1 _u_n_t_i_l s1 _d_o
         _f_o_r q := 1 _s_t_e_p 1 _u_n_t_i_l ncon _d_o
         _b_e_g_i_n
            p := p+1;
            cx[p] := (s−1) ×bs + bs_:2 + (RANDOM(bs) − bs_:2)_:spr;
            cy[p] := (r−1) ×hs + hs_:2 + (RANDOM(hs) − hs_:2)_:spr;
            fh[p] := fhmin + RANDOM(fhmax−fhmin) − 1;
            _i_f fh[p] < 1 _t_h_e_n fh[p] := 1;
            fv[p] := fvmin + RANDOM(fvmax−fvmin) − 1;
            _i_f fv[p] < 1 _t_h_e_n fv[p] := 1;
            v := 3.14159265/180 ×(amin + RANDOM(amax−amin));
            cosv[p] := cos(v);
            sinv[p] := sin(v)
         _e_n_d for q, s, and r;
         SHIFT(h+3);
         LINE;
         skrvtegn(62);
         red := _f_a_l_s_e;
         _f_o_r r := 1 _s_t_e_p 1 _u_n_t_i_l h _d_o



         _b_e_g_i_n
            _f_o_r s := 1 _s_t_e_p 1 _u_n_t_i_l b _d_o
            _b_e_g_i_n
               dmin := 1 105;
               _f_o_r p := 1 _s_t_e_p 1 _u_n_t_i_l p2 _d_o
               _b_e_g_i_n
                  e := cx[p];
                  f := cy[p];
                  g := cosv[p];
                  j := sinv[p];
                  d := (((e−s) ×g − (f−r) ×j)/fv[p])| ∧ 2
                       + (((e−s) ×j + (f−r) ×g)/fh[p])| ∧ 2;
                  _i_f d < dmin _t_h_e_n
                  _b_e_g_i_n
                     dmin := d;
                     p3 := p
                  _e_n_d if lower distance
               _e_n_d for p;
               p3 := p3 − 1;
               p3 := p3 − p3 _: 35 ×35 + 1;
               even := p3 = p3 _: 2 ×2;
               _i_f p3 > 9 _t_h_e_n
               p3 := _i_f p3 < 19 _t_h_e_n p3 + 39
               _e_l_s_e
               _i_f p3 < 28 _t_h_e_n p3 + 14 _e_l_s_e p3 − 10;
               _i_f even _= red _t_h_e_n
               _b_e_g_i_n
                  skrvtegn(_i_f red _t_h_e_n 62 _e_l_s_e 29);
                  red := −, red
               _e_n_d;
               skrvtegn(p3)
            _e_n_d for s;
            LINE
         _e_n_d for r
      _e_n_d inner MAP−block;
      LINE;  LINE; LINE;
      SHIFT(4);
      skrvtegn(62);
      skrvtekst(|<<Hvis De ønsker flere kort, saa skriv et 1−tal her: |>);
      r := tasttegn;
      _g_o_ _t_o _i_f r = 1 _t_h_e_n LL _e_l_s_e RESTART
   _e_n_d of MAP−block;
LINEQ:
   _b_e_g_i_n _c_o_m_m_e_n_t LINEQ−block;
      _i_n_t_e_g_e_r N, i, j;
      SHIFT(10);
      skrvtekst(|<<Type 3.  Løsning af tilfældige lineære ligninger.|>);
      LINE;  LINE;
      _i_f newlineq _t_h_e_n
      _b_e_g_i_n
         SHIFT(5);
         skrvtekst(|<<
     Programmet genererer  og løser et sæt af N tilfældige lineære ligninger.
Maximumværdien af N er 24.  Regnetiden for N = 20 er 23 sec.
i ALGOL og 4 sec. i maskinsprog.  Opgiv N = 0 for stop.
|>);
         linerest := linerest − 4;
         newlineq := _f_a_l_s_e
      _e_n_d if newlineq;
MM:   LINE;
      SHIFT(4);
      skrvtekst(|<<Opgiv N: |>);
      N := tast;
      LINE;
      _i_f N > 24 _t_h_e_n N := 24;
      _i_f N = 0 _t_h_e_n _g_o_ _t_o RESTART;
      _b_e_g_i_n _c_o_m_m_e_n_t inner LINEQ−block;



         _a_r_r_a_y x[1:N], MATRIX[1:N, 1:N + 1];
         _p_r_o_c_e_d_u_r_e  LINEQ1 (N, a, x, NOSOLUTION);
         _i_n_t_e_g_e_r  N;
         _a_r_r_a_y  a, x;
         _l_a_b_e_l  NOSOLUTION;
         _b_e_g_i_n
            _i_n_t_e_g_e_r p, i, j;
            _r_e_a_l  M;
            _f_o_r  p := 1 _s_t_e_p 1 _u_n_t_i_l N − 1 _d_o
            _b_e_g_i_n
               _f_o_r  i := p + 1 _s_t_e_p 1 _u_n_t_i_l N _d_o
               _b_e_g_i_n
                  _i_f  a[p,p] |= 0 _t_h_e_n _g_o_ _t_o L2;
                  _i_f  a[i,p] |= 0 _t_h_e_n _g_o_ _t_o L1;
                  _i_f  i < N _t_h_e_n _g_o_ _t_o L3;
                  _g_o_ _t_o NOSOLUTION;
L1:               _f_o_r j := p _s_t_e_p 1 _u_n_t_i_l N + 1 _d_o
                  _b_e_g_i_n
                     M := a[p,j];
                     a[p,j] := a[i,j];
                     a[i,j] := M
                  _e_n_d of row exchange;
                  _g_o_ _t_o L3;
L2:               _i_f a[i,p] = 0 _t_h_e_n _g_o_ _t_o L3;
                  M := −a[i,p]/a[p,p];
                  _f_o_r j := p+1 _s_t_e_p 1 _u_n_t_i_l N+1 _d_o
                  a[i,j] := a[i,j] + M ×a[p,j];
L3:            _e_n_d for i;
            _e_n_d for p;
            _i_f a[N,N] = 0 _t_h_e_n _g_o_ _t_o NOSOLUTION;
            _f_o_r p := N _s_t_e_p −1 _u_n_t_i_l 1 _d_o
            _b_e_g_i_n
               x[p] := a[p,N+1] := a[p,N+1]/a[p,p];
               _i_f p = 1 _t_h_e_n _g_o_ _t_o L4;
               _f_o_r i := p−1 _s_t_e_p −1 _u_n_t_i_l 1 _d_o
               a[i,N+1] := a[i,N+1] − x[p] ×a[i,p]
            _e_n_d for second p;
L4:      _e_n_d LINEQ−1;
NN:      _f_o_r i := 1 _s_t_e_p 1 _u_n_t_i_l N _d_o
         _f_o_r j := 1 _s_t_e_p 1 _u_n_t_i_l N+1 _d_o
         MATRIX[i,j] := RANDOM(30000);
         LINEQ1(N, MATRIX, x, ERROR);
         _g_o_ _t_o MM;
ERROR:   SHIFT(4);  LINE;
         skrvtekst(|<<Undskyld, determinanten er nul.  Her er et andet eksempel.|>);
         LINE;
         _g_o_ _t_o NN
      _e_n_d of inner LINEQ−block
   _e_n_d of LINEQ−block;
PRIME:
   _b_e_g_i_n _c_o_m_m_e_n_t prime block;
      _b_o_o_l_e_a_n first, last, small;  
      _i_n_t_e_g_e_r type, num, num1, fact, count, A, B;
      _i_n_t_e_g_e_r _p_r_o_c_e_d_u_r_e PRIM1(x);
      _i_n_t_e_g_e_r x;
      _b_e_g_i_n
         _i_n_t_e_g_e_r y;
A:       PRIM1 := x := x + 2;
         y := 1;
         _f_o_r y := y + 2 _w_h_i_l_e y ×y_<x _d_o
         _i_f (x_:y) ×y = x _t_h_e_n _g_o_ _t_o A
      _e_n_d;
      _p_r_o_c_e_d_u_r_e READ(number, text);
      _i_n_t_e_g_e_r number;
      _s_t_r_i_n_g text;
      _b_e_g_i_n
         _r_e_a_l N;



PP:      SHIFT(4);
         skrvtekst(text);
         N := tast;
         LINE;
         _i_f N < 1 ∨  N > 536870911 _t_h_e_n _g_o_ _t_o PP;
         number := N
      _e_n_d READ;
      SHIFT(10);
      skrvtekst(|<<Type 4.  Beregning af primtal.|>);
      LINE;  LINE;
      _i_f newprime _t_h_e_n
      _b_e_g_i_n
         SHIFT(8);
         skrvtekst(|<<
     Programmet indeholder to beregningstyper:
     1.  Beregning af primfaktorer i et opgivet tal, N.
     2.  Beregning af primtal i et opgivet interval fra A til B.
     Opgiv type 3 for stop.  Øvre grænse for tallene er 536870911.
|>);
         linerest := linerest − 5;
         newprime := _f_a_l_s_e
      _e_n_d of newprime;
OO:   LINE;
      READ(type, |<<Opgiv primtalberegningstype: |>);
      _i_f type = 1 _t_h_e_n
      _b_e_g_i_n
         READ(num, |<<Opgiv tallet, N: |>);
         num1 := num;
         first := _t_r_u_e;
         last := _f_a_l_s_e;
         _i_f num < 4 _t_h_e_n
QQ:      _b_e_g_i_n
            skrvtekst(|<<Primtal|>);
            _g_o_ _t_o OO
         _e_n_d;
         _f_o_r fact := 2, 3, PRIM1(fact) _w_h_i_l_e fact ×fact _< num1 ∧  num > 1 _d_o
         _b_e_g_i_n
            count := 0;
RR:         _i_f num_:fact ×fact = num _t_h_e_n
SS:         _b_e_g_i_n
               count := count + 1;  
               num := num_:fact;
               _i_f first _t_h_e_n skrvtekst(|<<=|>);
               _i_f count = 1 _t_h_e_n
               _b_e_g_i_n
                  _i_f −, first _t_h_e_n skrvtekst(|<< ×|>);  
                  first := _f_a_l_s_e;
                  skrv(_i_f fact < 10 _t_h_e_n |<d|>
                  _e_l_s_e _i_f fact < 100 _t_h_e_n |<dd|>
                  _e_l_s_e _i_f fact < 1000 _t_h_e_n |<ddd|>
                  _e_l_s_e _i_f fact < 10000 _t_h_e_n |<dddd|>
                  _e_l_s_e |<ddddddddd|>, fact)
               _e_n_d if count = 1;
               _g_o_ _t_o _i_f last _t_h_e_n OO _e_l_s_e RR
            _e_n_d if divisor;
            _i_f count > 1 _t_h_e_n
            _b_e_g_i_n
               skrvtekst(|<<| ∧ |>);
               skrv(_i_f count < 10 _t_h_e_n |<d|> _e_l_s_e |<dd|>, count)
            _e_n_d if power printing;
            count := 0
         _e_n_d for fact;
         _i_f first _t_h_e_n _g_o_ _t_o QQ;
         last := _t_r_u_e; 
         fact := num;
         _i_f num > 1 _t_h_e_n _g_o_ _t_o SS;
         _g_o_ _t_o OO



      _e_n_d if type = 1
      _e_l_s_e
      _i_f type = 2 _t_h_e_n
      _b_e_g_i_n
         READ(A, |<<Opgiv nedre grænse, A: |>);
         READ(B, |<<Opgiv øvre grænse, B: |>);
         small := B < 10000;
         count := 0;
         fact := _i_f small _t_h_e_n 10 _e_l_s_e 8;
         SHIFT(4);
         _i_f A = 1 _t_h_e_n A := 2
         _e_l_s_e
         _i_f A > 3 _t_h_e_n
         _b_e_g_i_n
            A := A − (_i_f A_:2 ×2 = A _t_h_e_n 1 _e_l_s_e 2);
UU:         A := PRIM1(A)
         _e_n_d;
TT:      _i_f count_:fact ×fact = count _t_h_e_n LINE;
         count := count + 1;  
         _i_f A_<B _t_h_e_n skrv(_i_f small _t_h_e_n |<−ddddd|> _e_l_s_e |<−ddddddddd|>, A);
         _i_f A < B _t_h_e_n
         _b_e_g_i_n
            _i_f A = 2 _t_h_e_n
            _b_e_g_i_n
               A := 3;
               _g_o_ _t_o TT
            _e_n_d;
            _g_o_ _t_o UU
         _e_n_d;
         _g_o_ _t_o OO
      _e_n_d type = 2
      _e_l_s_e
      _g_o_ _t_o RESTART
   _e_n_d prime block;
LANU: 
   _b_e_g_i_n _c_o_m_m_e_n_t large number calculation block;
      _b_o_o_l_e_a_n first, out;
      _i_n_t_e_g_e_r M, carry, count, c1, c2, d1, d2, type, N, alimit, asize, nn, a,
              b, D, bsize, m;
      _p_r_o_c_e_d_u_r_e READ(number, text);
      _i_n_t_e_g_e_r number;
      _s_t_r_i_n_g text;
      _b_e_g_i_n
         SHIFT(4);
         skrvtekst(text);
         number := tast;
         LINE
      _e_n_d READ;
      _p_r_o_c_e_d_u_r_e ALARM(n);
      _v_a_l_u_e n;
      _i_n_t_e_g_e_r n;
      skrvtekst(|<<
FEJL |>, _i_f n = 1 _t_h_e_n |<<1|> _e_l_s_e |<<2|>);
      _p_r_o_c_e_d_u_r_e MULT(n, A, size);
      _v_a_l_u_e n;
      _i_n_t_e_g_e_r n, size;
      _i_n_t_e_g_e_r _a_r_r_a_y A;
      _b_e_g_i_n
         carry := 0;
         _f_o_r count := 0 _s_t_e_p 1 _u_n_t_i_l alimit _d_o
         _b_e_g_i_n
            c1 := A[count];
            c2 := c1_:M;
            c1 := (c1−c2 ×M)×n + carry;
            carry := c1_:M;
            c1 := c1 − carry ×M;
            c2 := c2 ×n + carry;



            carry := c2_:M;
            A[count] := (c2−carry ×M)×M + c1;
            _i_f count = size _t_h_e_n
            _b_e_g_i_n
               _i_f carry = 0 _t_h_e_n _g_o_ _t_o ex
               _e_l_s_e
               _i_f count < alimit _t_h_e_n size := size + 1
               _e_l_s_e
               ALARM(1)
            _e_n_d if count
         _e_n_d for count;
ex:   _e_n_d MULT;
      _p_r_o_c_e_d_u_r_e DIV(n, A, size, empty);
      _v_a_l_u_e n;
      _i_n_t_e_g_e_r n, size;
      _b_o_o_l_e_a_n empty;
      _i_n_t_e_g_e_r _a_r_r_a_y A;
      _b_e_g_i_n
         first := _t_r_u_e;
         carry := 0;
         _f_o_r count := size _s_t_e_p −1 _u_n_t_i_l 0 _d_o
         _b_e_g_i_n
            c1 := A[count];
            c2 := c1_:M;
            c1 := c1 − c2 ×M;
            carry := carry ×M + c2;
            c2 := carry_:n;
            carry := (carry − c2 ×n) ×M + c1;
            c1 := carry_:n;
            carry := carry − c1 ×n;
            A[count] := c1 := c1 + c2 ×M;
            _i_f first _t_h_e_n
            _b_e_g_i_n
               _i_f c1 > 0 _t_h_e_n first := _f_a_l_s_e
               _e_l_s_e
               _i_f size > 0 _t_h_e_n size := size − 1
            _e_n_d if first
         _e_n_d for count;
         empty := first ∧  c1 = 0
      _e_n_d DIV;
      _p_r_o_c_e_d_u_r_e ADD(plus, A, B, asize, bsize);
      _v_a_l_u_e plus, bsize;
      _b_o_o_l_e_a_n plus;
      _i_n_t_e_g_e_r asize, bsize;
      _i_n_t_e_g_e_r _a_r_r_a_y A, B;
      _b_e_g_i_n
         carry := 0;
         _f_o_r count := 0 _s_t_e_p 1 _u_n_t_i_l alimit _d_o
         _b_e_g_i_n
            c1 := A[count];  d1 := B[count];
            c2 := c1_:M;      d2 := d1_:M;
            c1 := c1−c2 ×M;   d1 := d1 − d2 ×M;
            c1 := c1 + (_i_f plus _t_h_e_n d1 _e_l_s_e − d1) + carry;
            carry := 0;
L1:         _i_f c1 < 0 _t_h_e_n
            _b_e_g_i_n
               c1 := c1 + M;
               carry := carry − 1;
               _g_o_ _t_o L1
            _e_n_d if c1 negative;
            d1 := c1_:M;
            c1 := c1 − d1 ×M;
            c2 := c2 + (_i_f plus _t_h_e_n d2 _e_l_s_e − d2) + d1 + carry;
            carry := 0;
L2:         _i_f c2 < 0 _t_h_e_n
            _b_e_g_i_n
               c2 := c2 + M;



               carry := carry − 1;
               _g_o_ _t_o L2
            _e_n_d if c2 negative;
            d1 := c2_:M;
            c2 := c2 − d1 ×M;
            carry := carry + d1;
            A[count] := c1 := c1 + c2 ×M;
            _i_f count _> bsize ∧  carry = 0 _t_h_e_n _g_o_ _t_o L3
         _e_n_d for count;
         _i_f carry |= 0 _t_h_e_n ALARM(2);
L3:      first := _t_r_u_e;
         _f_o_r count := alimit _s_t_e_p −1 _u_n_t_i_l 0 _d_o
         _b_e_g_i_n
            asize := count;
            _i_f A[count] |= 0 _t_h_e_n _g_o_ _t_o L4
         _e_n_d;
L4:   _e_n_d ADD;
      _p_r_o_c_e_d_u_r_e P4(n);
      _v_a_l_u_e n;
      _i_n_t_e_g_e_r n;
      _b_e_g_i_n
         _i_n_t_e_g_e_r i, z, D, a;
         D := 1000;
         z := _i_f first _t_h_e_n 0 _e_l_s_e 16;
         _f_o_r i := 1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
         _b_e_g_i_n
            a := n_:D;
            n := n − a ×D;
            _i_f a |= 0 _t_h_e_n
            _b_e_g_i_n
               skrvtegn(a);
               first := _f_a_l_s_e;
               z := 16
            _e_n_d
            _e_l_s_e skrvtegn(z);
            D := D_:10
         _e_n_d for i
      _e_n_d P4;
      _p_r_o_c_e_d_u_r_e PR(A, size);
      _v_a_l_u_e size;
      _i_n_t_e_g_e_r size;
      _a_r_r_a_y A;
      _b_e_g_i_n
         first := _t_r_u_e;
         d1 := 0;
         _f_o_r count := size _s_t_e_p −1 _u_n_t_i_l 0 _d_o
         _b_e_g_i_n
            c1 := A[count];
            c2 := c1_:M;
            c1 := c1 − c2 ×M;
            P4(c2);
            skrvml(1);
            P4(c1);
            skrvml(1);
            d1 := d1 + 1;
            _i_f d1_:8 ×8 = d1 _t_h_e_n LINE
         _e_n_d for count
      _e_n_d PR;
      SHIFT(10);
      skrvtekst(|<<Type 5.  Beregning af store tal.|>);
      LINE;  LINE;
      _i_f newlanu _t_h_e_n
      _b_e_g_i_n
         SHIFT(8);
         skrvtekst(|<<
        Programmet indeholder fire beregningstyper:
        1. Beregning af fakultet: FAC(N) = 1 ×2×3×4........ ×N.



        2. Beregning af potens: a| ∧ b.
        3. Beregning af e = 2.718 ...... med D cifre.
        4. Beregning af pi = 3.1415.... med D cifre.
        5. giver programstop.
|>);
         linerest := linerest − 7;
         newlanu := _f_a_l_s_e
      _e_n_d of newlanu;
      LINE;
      M := 10000;
VV:   READ(type, |<<Opgiv beregningstype for store tal: |>);
      _i_f type = 1 _t_h_e_n
      _b_e_g_i_n
         READ(N, |<<Opgiv N:|>);
         _i_f N > 1000 _t_h_e_n N := 1000;
         alimit := 0.05 ×N×ln(N);
         _b_e_g_i_n
            _i_n_t_e_g_e_r _a_r_r_a_y FAC[0:alimit];
            _f_o_r count := 0 _s_t_e_p 1 _u_n_t_i_l alimit _d_o FAC[count] := 0;
            asize := 0;
            FAC[0] := 1;
            _f_o_r nn := 1 _s_t_e_p 1 _u_n_t_i_l N _d_o MULT(nn, FAC, asize);
            LINE;  LINE;
            SHIFT(4);
            skrvtekst(|<<FAC :=|>);
            LINE;
            PR(FAC, asize);
            LINE;
            _g_o_ _t_o VV
         _e_n_d block
      _e_n_d if type = 1
      _e_l_s_e
      _i_f type = 2 _t_h_e_n
      _b_e_g_i_n
         READ(a, |<<Opgiv a: |>);
         READ(b, |<<Opgiv b: |>);
         alimit := 1 + 0.055 ×b×ln(a);
         _b_e_g_i_n
            _i_n_t_e_g_e_r _a_r_r_a_y POT[0:alimit];
            _f_o_r count := 0 _s_t_e_p 1 _u_n_t_i_l alimit _d_o POT[count] := 0;
            asize := 0;
            POT[0] := 1;
            _f_o_r nn := 1 _s_t_e_p 1 _u_n_t_i_l b _d_o MULT(a, POT, asize);
            LINE;  LINE;
            SHIFT(4);
            skrvtekst(|<<a| ∧ b := |>);
            LINE;
            PR(POT, asize);
            LINE;
            _g_o_ _t_o VV
         _e_n_d block
      _e_n_d if type = 2
      _e_l_s_e
      _i_f type < 5 _t_h_e_n
      _b_e_g_i_n
         READ(D, |<<Opgiv D: |>);
         alimit := D_:8;
         _i_f alimit ×8 |= D _t_h_e_n
         _b_e_g_i_n
            skrvtekst(|<<D er ændret til:|>);
            alimit := alimit + 1;
            D := 8 ×alimit;
            skrv(|<nddd|>, D);
            LINE
         _e_n_d;
         D := 8 ×alimit
      _e_n_d



      _e_l_s_e _g_o_ _t_o RESTART;
      _i_f type = 3 _t_h_e_n
      _b_e_g_i_n
         _i_n_t_e_g_e_r _a_r_r_a_y RESULT, TERM[0:alimit];
         _f_o_r count := 0 _s_t_e_p 1 _u_n_t_i_l alimit _d_o
         RESULT[count] := TERM[count] := 0;
         asize := bsize := alimit;
         RESULT[alimit] := 2;
         TERM[alimit] := 1;
         out := _f_a_l_s_e;
         m := 1;
         _f_o_r m := m + 1 _w_h_i_l_e −, out _d_o
         _b_e_g_i_n
            DIV(m, TERM, bsize, out);
            ADD(_t_r_u_e, RESULT, TERM, asize, bsize)
         _e_n_d for m;
         LINE;  LINE;
         SHIFT(4);
         skrvtekst(|<<e ×10|∧ D := |>);
         LINE;
         PR(RESULT, asize);
         LINE;
         _g_o_ _t_o VV
      _e_n_d block if type = 3
      _e_l_s_e
      _b_e_g_i_n
         _b_o_o_l_e_a_n out1, out2, out3, plus;
         _i_n_t_e_g_e_r t1size, t2size, t3size, ssize;
         _i_n_t_e_g_e_r _a_r_r_a_y RESULT, T1, T2, T3, SUM[0:alimit];
         _f_o_r count := 0 _s_t_e_p 1 _u_n_t_i_l alimit _d_o
         RESULT[count] := T1[count] := T2[count] := T3[count] := 0;
         T1[alimit] := T2[alimit] := T3[alimit] := 24;
         asize := t1size := t2size := t3size := alimit;
         DIV(8, T1, t1size, out1);
         DIV(171, T2, t2size, out2);
         DIV(1434, T3, t3size, out3);
         plus := _f_a_l_s_e;
         m := −1;
         _f_o_r m := m + 2 _w_h_i_l_e −, out1 _d_o
         _b_e_g_i_n
            _f_o_r count := 0 _s_t_e_p 1 _u_n_t_i_l alimit _d_o SUM[count] := 0;
            ssize := 0;
            ADD(_t_r_u_e, SUM, T1, ssize, t1size);
            _i_f −, out2 _t_h_e_n ADD(_t_r_u_e, SUM, T2, ssize, t2size);
            _i_f −, out3 _t_h_e_n ADD(_t_r_u_e, SUM, T3, ssize, t3size);
            DIV(m, SUM, ssize, out);
            plus := −, plus;
            ADD(plus, RESULT, SUM, asize, ssize);
            DIV(64, T1, t1size, out1);
            _i_f −, out2 _t_h_e_n DIV(3249, T2, t2size, out2);
            _i_f −, out3 _t_h_e_n _f_o_r nn := 1,2 _d_o DIV(239, T3, t3size, out3)
         _e_n_d for m;
         LINE;  LINE;
         SHIFT(4);
         skrvtekst(|<<pi ×10|∧ D :=|>);
         LINE;
         PR(RESULT, asize);
         LINE;
         _g_o_ _t_o VV
      _e_n_d type = 4
   _e_n_d LANU−block;
FINISH:
_e_n_d of program DEMON−1D;


